

# How to Visualize the Dance of the Money Bees Using Animated Graphs in SAS/INSIGHT

Mang-King W. Cheung, Kaiser Permanente, Los Angeles, CA

## INTRODUCTION

In the investment book, *Dance of the Money Bees* (1974), John Train described the stock market swarming up-and-down just like the dance of the honeybees. What an exquisite way of depicting the volatility of the stock market! The goal of this paper is to demonstrate a way to visualize the dance using animated graphs in SAS INSIGHT®.

In order to reach the goal, several steps are involved:

1. Read data from a Microsoft® Excel spreadsheet using PROC ACCESS.
2. Write a SAS® MACRO to compute the returns and risks.
3. Form the final data structure for SAS INSIGHT®.
4. Visualize the dance using animated graphs in SAS INSIGHT®.

## STEP 1: READ DATA FROM AN EXCEL SPREADSHEET USING PROC ACCESS

There are several ways to read data from an Excel spreadsheet into the SAS system. One of the more efficient and automated ways is to use PROC ACCESS with DBMS=EXCEL.

Source code to accomplish the task:

```
PROC ACCESS DBMS=excel;
CREATE work._imex_.access;
PATH='c:\beedance\daily2k.xls';
WORKSHEET='sasdata';
GETNAMES=yes;
SCANTYPE=yes;

MIXED=yes;
CREATE work._imex_.view;
SELECT all;
RUN;

DATA work.nav;
SET work._imex_;
RUN;
```

The data structure of my Excel spreadsheet is shown in Table 1. The spreadsheet stores daily closing net asset values of a series of sector mutual funds. The first column is the trading date followed by columns of net asset values of different mutual funds.

Table 1: Data Structure of the Excel Spreadsheet

	A	N	O	P	Q	R
1	tdate	bio	bmm	comp	food	lei
639	12/18/00	88.78	39.45	55.49	45.83	61.55
640	12/19/00	86.84	39.34	52.96	46.16	61.25
641	12/20/00	81.04	37.55	48.48	46.01	58.41
642	12/21/00	80.54	37.74	48.23	46.27	58.76
643	12/22/00	85.33	38.57	53.84	46.25	59.99
644	12/26/00	86.11	38.33	52.62	46.62	59.59
645	12/27/00	87.34	38.48	54.10	47.09	60.06
646	12/28/00	87.95	39.11	54.68	47.46	61.08
647	12/29/00	86.80	39.32	52.33	47.64	61.45

Fidelity Investments®, a well-known investment firm in the United States, offers almost 40 different sector funds for the public to invest. Specialized funds shown in Table 1 include biotechnology

(bio), multimedia (bmm), computers (comp), food and agriculture (food), leisure (lei). A complete listing of the sector funds that I continually track is shown in the Appendix. Tracking these sector funds provides an excellent barometer of the stock market.

## STEP 2: WRITE A SAS® MACRO TO COMPUTE THE RETURNS AND RISKS

The purpose of this step is to convert the data structure in Table 1 to a new database that contains information about returns and risks for each mutual fund in Step 1. The data elements in this return-and-risk database are shown in Table 2. The investment return is defined by the percentage change in the net asset values between two selected dates. The investment risk is related to the standard deviation of daily changes in the net asset values between two selected dates. The periods that I chose for the selected dates include: 5 weeks (about 1 month), 13 weeks (3 months), 26 weeks (6 months), and 52 weeks (1 year). The selection of the period lengths is arbitrary. You could tailor the time periods to your own preference.

Table 2: Data Structure of the Return and Risk Data Set

	Int	Nom	Int	Int	Int	Int
52	dateval	mfund	retnw5	riskw5	retnw13	riskw13
12	12/29/2000	bio	1.8899	3.7282	-11.8961	3.4812
13	12/29/2000	bmm	-0.7572	1.7284	-13.6255	1.7757
14	12/29/2000	comp	-18.2259	5.0371	-36.1366	4.4507
15	12/29/2000	food	5.6978	0.8289	17.2900	0.7568
16	12/29/2000	lei	-1.1263	1.6624	-13.0342	1.7511
17	12/29/2000	fins	13.2866	1.4859	5.7757	1.7002
18	12/29/2000	gold	14.9325	1.6369	3.0868	1.3440
19	12/29/2000	atran	3.2838	1.5356	10.4017	1.4567
20	12/29/2000	auto	0.0507	1.0804	-3.7109	1.1170
21	12/29/2000	indue	7.6391	2.0666	0.1760	1.8574
22	12/29/2000	chm	12.7202	1.6686	21.5878	1.6115

The data columns in Table 2 are defined as follows:

DATEVAL	Date
MFUND	Label of the mutual fund (see Appendix). You could also create your own labels and names.
RETNW5	Percentage return in 5 weeks
RISKW5	Standard deviation of log ratios between two consecutive daily prices in 5 weeks
RETNW13	Percentage return in 13 weeks
RISKW13	Standard deviation of log ratios between two consecutive daily prices in 13 weeks

We need to transform the data in Table 1 (where the rows are dates and columns are mutual funds) to Table 2 (where the rows are mutual funds and the columns are returns and risks).

This data transformation process involves two major steps:

1. A PROC MEANS procedure is required to compute means (returns) and standard deviations (risks).
2. A PROC TRANSPOSE procedure is needed to convert the mutual funds from column variables to row observations.

The following is the logic of a MACRO procedure and its source code to compute the returns and risks for a selected length of time. The MACRO contains only one parameter, which is the length of time measured by the number of weeks.

Before running the MACRO, I specify two global MACRO variables: &NOWDATE and &FVAR. &NOWDATE could be any

trading date. As for the &FVAR variable, I labeled 37 mutual funds that I am tracking: bio, bmm, ..., meq. (See Appendix for the complete listing of funds and associated labels.) You could track different collection of mutual funds, stocks or other equities, and create your own labels that are meaningful to you.

```
%LET nowdate=29dec2000;
%LET fvar=
bio  bmm  comp food lei  fins gold  atran
auto  indue chm  compd softc defa devcm indum
elect ener  eners medi conhs bkg  envi  ngas
ins   paper hlth tech rbank tran  retail telcm
util  homef real buss meq
;
```

Then the MACRO step is defined by the following code:

```
%MACRO weeks(i);
DATA nav&i;
SET nav
;
FORMAT dateval MMDDYY10.;
dateval="&nowdate"d;
weekback=&i;

IF (dateval-7*weekback) <= tdate <= dateval;

ARRAY f{37} &fvar;
DO j=1 TO 37;
  f{j}=LOG(f{j})-LOG(LAG(f{j}));
END;

PROC MEANS DATA=nav&i SUM STD NWAY NOPRINT;
VAR &fvar
;
OUTPUT OUT=new&i(KEEP=&fvar)
;
PROC TRANSPOSE DATA=new&i
  OUT=newtr&i(KEEP=_NAME_ COL1 COL4 COL5
    RENAME=( _NAME_=mfund COL1=n COL4=mean
      COL5=risk));
RUN;

DATA weeks&i;
KEEP mfund retnw&i riskw&i;
SET newtr&i;
retnw&i=100*(EXP(n*mean)-1);
riskw&i=100*risk;

PROC DATASETS LIBRARY=work MEMTYPE=DATA
  NOLIST;
DELETE nav&i new&i newtr&i;
QUIT;
%MEND weeks;
```

A few points need to be further explained for the above MACRO logic:

1. IF (DATEVAL-7\*WEEKBACK) <= TDATE <= DATEVAL;

This selects a subset of data from Table 1. The subset depends on how many weeks of data we specify in the MACRO. If we want to compute returns and risks for 5 weeks, then &i=5 and WEEKBACK=5. This would select data with TDATE between DATEVAL and DATEVAL-35; and DATEVAL is specified by &NOWDATE.

2. F{J}=LOG(F{J})-LOG(LAG(F{J}));

In essence, F{J} is the log ratio of two consecutive daily prices. The PROC MEANS procedure would produce the SUM and STD for the log ratios. The STD of the log ratios is used to define RISK. The SUM of the log ratios would cancel out all in-between terms and return the log ratio of the first and last net asset values in the data subset. Taking the exponential function of the log ratio

yields the ratio of the first and the last net asset values in the data subset. Thus,

$$\text{RETNW5}=100*(\text{EXP}(N*\text{MEAN})-1)$$

yields the percentage return for 5 weeks.

### STEP 3: FORM THE FINAL DATA STRUCTURE FOR SAS INSIGHT®

With the MACRO WEEKS in Step 2 defined, we could run the MACRO several times to get the returns and risks data for 5 week, 13 weeks, 26 weeks, and 52 weeks. Then we can combine the data using a DATA step and a MERGE statement to form the data set that I named as ONEDAY.

The source code are shown as follows:

```
%weeks(5)
%weeks(13)
%weeks(26)
%weeks(52)

DATA oneday;
dateval="&nowdate"D;
MERGE weeks5 weeks13 weeks26 weeks52;
FORMAT dateval MMDDYY10.;
RUN;
PROC DATASETS LIBRARY=work MEMTYPE=DATA
  NOLIST;
DELETE weeks5 weeks13 weeks26 weeks52;
QUIT;
```

Now we have transformed the data from Table 1 to Table 2. Let's take a look on the data set ONEDAY.

```
PROC INSIGHT DATA=oneday;
  SCATTER retnw5 retnw13 retnw26 retnw52 *
    riskw52 / LABEL=mfund;


RUN;
```

Figure 1 (on the following page) shows the scatter plot matrix of RETNW5, RETNW13, RETNW26, and RETNW52 versus RISKW52. From Figure 1, you could simultaneously view four RETN versus RISK charts. With these charts, you could evaluate the performance of different mutual funds in multiple time frames. Many investment textbooks show examples of these Return-Risk plots when they discuss portfolio selection and the capital asset pricing model.

The data set ONEDAY shows only one day of reference on returns and risks. When you repeat the MACRO WEEKS procedure for multiple dates and append the data together, you form the final data structure (Table 3) for SAS INSIGHT® to run the animation. Table 3 is very similar to Table 2, except that Table 3 contains multiple dates of information on returns and risks.

Table 3: Final Data Structure for SAS INSIGHT® Animation

	Int	Non	Int	Int	Int	Int
740	dateval	mfund	retnw5	riskw5	retnw13	riskw13
698	12/28/2000	telcm	-7.4135	2.9187	-24.8443	2.8445
699	12/28/2000	util	-3.8126	1.8620	-18.9870	1.7725
700	12/28/2000	homef	21.7833	1.3511	21.0766	1.5614
701	12/28/2000	real	9.7054	0.6214	6.4626	0.6532
702	12/28/2000	buss	2.5321	1.4913	6.5927	1.3294
703	12/28/2000	meq	6.6997	1.1501	6.0661	1.1332
704	12/29/2000	bio	1.8899	3.7282	-11.8961	3.4812
705	12/29/2000	bmm	-0.7572	1.7284	-13.6255	1.7757
706	12/29/2000	comp	-18.2259	5.0371	-36.1366	4.4507
707	12/29/2000	food	5.6978	0.8289	17.2900	0.7568
708	12/29/2000	lei	-1.1263	1.6624	-13.0342	1.7511
709	12/29/2000	fins	13.2866	1.4859	5.7757	1.7002
710	12/29/2000	gold	14.9325	1.6369	3.0868	1.3440
711	12/29/2000	atran	3.2838	1.5356	10.4017	1.4567



Current Date: 29DEC2000

Figure 1: Scatter plot Matrix of RETN5, RETN13, RETN26, RETN52 versus RISKW52

### STEP 4: VISUALIZE THE DANCE USING ANIMATED GRAPHS IN SAS INSIGHT®

With the data in Table 3 available, you could run PROC INSIGHT and get a scatter plot of return versus risk for all dates in the data set. Inside the EDIT drop-down menu, first choose the Windows/ Animate option (Figure 2). Then using the Animation Dialog in SAS INSIGHT® (Figure 3), you can refresh the scatter plot of return versus risk with one day of data at a time. The resulting visual effect would be just like the dance of a swarm of bees with each bee representing one mutual fund in the data set.

Figure 2: EDIT Drop-Down Menu





Figure 3: Animation Dialog in SAS INSIGHT®



### CONCLUSION

The goal of this paper is to demonstrate a way to visualize the dance of the money bees using animated graphs in SAS INSIGHT®. In his book, *Dance of the Money Bees*, John Train wrote:

“When Von Frisch studied how a honeybee tells his mates the location of a treasure of blossoms, he discovered that the news is transmitted through a dance.”

Now we have a tool to observe the dance of the money bees. Can we decipher the signals and locate the treasure of blossoms?

### APPENDIX: LIST OF SECTOR FUNDS

Name of Sector Fund	Label
Air Transportation	atran
Automotive	auto
Banking	rbank
Biotechnology	bio
Brokerage & Investment Management	bkg
Business Services & Outsourcing	buss
Chemicals	chm
Computers	comp
Construction & Housing	conhs
Consumer Industries	conpd
Defense & Aerospace	defa
Developing Communications	devcm
Electronics	elect
Energy	ener
Energy Service	eners
Environmental Services	envi
Financial Services	fins
Food & Agriculture	food
Gold	gold
Health Care	hith
Home Finance	homef
Industrial Equipment	indue
Industrial Materials	indum
Insurance	ins
Leisure	lei
Medical Delivery	medi
Medical Equipment and Systems	meq
Multimedia	bmm
Natural Gas	ngas
Paper & Forest Products	paper
Real Estate	real
Retailing	retail
Software & Computer Services	softc
Technology	tech
Telecommunications	telcm
Transportation	tran
Utilities Growth	util

### REFERENCES

- Bodie, Zvi and Alex Kane, Alan Marcus (1996), *Investment*, Third edition, Irwin.
- Train, John (1974), *Dance of the Money Bees*, Harper & Row, Publishers.
- SAS Institute Inc. (1996), *SAS/INSIGHT User's Guide*, Version 6, Third edition.

### ACKNOWLEDGMENTS

I would like to thank Jay Gravink and Mae-Fay Yeh for proofreading and editing this paper. Without their effectual and supportive comments, this manuscript would be filled with problems of grammar and usage. Of course, I am still responsible for any existing blunders.

**TRADEMARK CITATION**

SAS, SAS/ACCESS, SAS/BASE, SAS/INSIGHT are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries.

Other brand and product names are registered trademarks or trademarks of their respective companies.

® indicates USA registration.

**CONTACT INFORMATION**

Your comments and questions are valued and encouraged.  
Contact the author at:

Mang-King W Cheung  
Kaiser Permanente  
393 East Walnut Street, 7/F CMS  
Pasadena, CA 91188  
Work Phone: (626)-405-6872  
Fax: (626)-405-6456  
Email: [Mang.K.Cheung@kp.org](mailto:Mang.K.Cheung@kp.org)